



GraphDB em docker

José Carlos Ramalho
2024-03-02

Imagem docker do graphDB



[Explore](#) / khaller/graphdb-free



khaller/graphdb-free ☆

By [khaller](#) • Updated 8 days ago

Docker images for the free version of GraphDB.

Image

Overview

Tags

Supported tags

- GraphDB 10: latest, 10.5.1, 1.5.1-graphdb10.5.1
- GraphDB 9: 9.11.2, 1.3.6-graphdb9.11.2

Supported architectures

- x64/amd64
- arm64

Docker Pull Command

```
docker pull khaller/graphdb-free
```

Copy

Execução

```
$ docker images
```

```
...
```

```
$ docker run -d -p 7200:7200 --name graphdb <id imagem>
```

```
$ docker stop graphdb
```

Parar o container e apagá-lo

```
$ docker rm graphdb
```

```
$ docker stop graphdb
```

Parar o container e rearrancá-lo

```
$ docker restart graphdb
```

Persistência de dados: volumes

```
$ docker volume create graphdb_data  
$ docker run  
    -d  
    -p 7200:7200  
    -v graphdb_data:/opt/graphdb/home  
    --name my_graphdb_instance  
    <id imagem>
```

Ficha de aula

1. Criar um volume de dados;
2. Colocar o container em execução com persistência de dados;
3. Criar um repositório de nome “tabelaPeriodica” e carregar o dataset dado pelo docente;
4. Navegar no repositório e verificar a sua estrutura: nº de triplos, conceitos, ...
5. Parar o container;
6. Colocar o container de novo em execução e verificar que o repositório continua lá.

SPARQL: exercícios

1. Especifica uma query SPARQL para responder à questão: Quantos elementos há na tabela periódica?
2. Especifica uma query SPARQL para responder à questão: Qual o símbolo, o nome e o número atômico de cada elemento?
3. Obtenha a última lista ordenada por número atômico.
4. Especifica uma query SPARQL para responder à questão: Quais são os 3 elementos com maior número atômico?
5. Cada elemento na tabela pertence a um grupo. Obtenha o número e o nome de cada grupo (há cerca de 20 grupos).
6. Obtenha o nome, o símbolo e o número atômico dos elementos do grupo 18.
7. Qual o número de elementos em cada grupo? (distribuição dos elementos por grupos)
8. Quais os 5 grupos com mais elementos?
9. Qual a média dos números atômicos dos elementos na tabela?
10. Qual o maior número atômico na tabela?
11. Qual o menor número atômico na tabela?
12. Qual o nome do elemento com esse número atômico?



Q1: Quantos elementos há na tabela periódica?

```
PREFIX : <http://www.daml.org/2003/01/periodictable/PeriodicTable#>

select (count (?s) as ?numero) where {
    ?s a :Element.
}
```

Q2: Qual o símbolo, o nome e o número atômico de cada elemento?

```
PREFIX : <http://www.daml.org/2003/01/periodictable/PeriodicTable#>

select ?n ?nome ?simb where {
    ?s a :Element .
    ?s :name ?nome.
    ?s :symbol ?simb.
    ?s :atomicNumber ?n.
}
order by ?n
```


Q4: Quais são os 3 elementos com maior número atômico?

```
PREFIX : <http://www.daml.org/2003/01/periodictable/PeriodicTable#>

select ?n ?nome ?simb where {
    ?s a :Element .
    ?s :name ?nome.
    ?s :symbol ?simb.
    ?s :atomicNumber ?n.
}
order by desc(?n)
limit 3
```

Q5: Obtenha o número e o nome de cada grupo (há cerca de 20 grupos)

```
PREFIX : <http://www.daml.org/2003/01/periodictable/PeriodicTable#>

select ?nome ?num where {
    ?s a :Group .
    optional {?s :name ?nome.}
    optional {?s :number ?num.}
}
```

Q6: Obtenha o nome, o símbolo e o número atômico dos elementos do grupo 18

```
PREFIX : <http://www.daml.org/2003/01/periodictable/PeriodicTable#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>

select ?n ?nome ?simb where {
    ?s a :Element .
    ?s :name ?nome.
    ?s :symbol ?simb.
    ?s :atomicNumber ?n.
    ?g a :Group.
    ?g :number "18"^^xsd:integer.
    ?s :group ?g
}
order by ?n
```

Q7: Qual o número de elementos em cada grupo? (distribuição dos elementos por grupos)

```
PREFIX : <http://www.daml.org/2003/01/periodictable/PeriodicTable#>

select ?grupo (count (?el) as ?conta) where {
    ?grupo a :Group.
    ?grupo :element ?el.
}
group by ?grupo
order by desc(?conta)
```

Q9: Qual a média dos números atômicos dos elementos na tabela?

```
PREFIX : <http://www.daml.org/2003/01/periodictable/PeriodicTable#>

select (avg (?na) as ?media) where {
    ?s a :Element.
    ?s :atomicNumber ?na
}
```

Q10: Qual o maior número atómico na tabela?

```
PREFIX : <http://www.daml.org/2003/01/periodictable/PeriodicTable#>

select (max (?na) as ?maximo) where {
    ?s a :Element.
    ?s :atomicNumber ?na
}
```